Добрый день
Возможно ли в запросе выбрать номенклатуру и 3 последних документа поступления для нее?
---
Ну дак сделайте запрос по РН Продажи отберите различную номенклатуру выгрузите в ТЗ, строчки номенклатуры из ТЗ перебирая циклом загоняйте в тот запрос что я выше написал.

Либо раскуривайте запросом тут последнее сообщение
---
да возможно, доки только сортируешь по дате максимум и выбираешь первые 3
---
ВЫБРАТЬ ПЕРВЫЕ 3 - предварительно упорядочить по дате документа
---
```1c
    ВЫБРАТЬ РАЗЛИЧНЫЕ ПЕРВЫЕ 3
    ТоварыНаСкладах.Регистратор КАК Регистратор,
    ТоварыНаСкладах.Период КАК Период
    ИЗ
    РегистрНакопления.ТоварыНаСкладах КАК ТоварыНаСкладах
    ГДЕ
    ТоварыНаСкладах.Номенклатура = &Номенклатура
    И ТИПЗНАЧЕНИЯ(ТоварыНаСкладах.Регистратор) = ТИП(Документ.ПоступлениеТоваров)

    СГРУППИРОВАТЬ ПО
    ТоварыНаСкладах.Регистратор,
    ТоварыНаСкладах.Период

    УПОРЯДОЧИТЬ ПО
    ТоварыНаСкладах.Период УБЫВ
    АВТОУПОРЯДОЧИВАНИЕ
```
---
надо ведь сначала запрос по Справочнику сделать Номенклатуры ?
---
По всей номенклатуре хотите получить 3 последних поступления? Сколько элементов в справочнике номенклатуры? Сдается мне это будет выполняться ну очень долго
---
по номенклатуре которая была в продаже за определенный период
И да список номенклатуры не маленький
---
ну тогда из товаров на складах, отбираешь сначала проданную номенклатура за период, группируешь, во второй таблице как в (4) и соединяешь.
---
Не понимаю как передать &Номенклатура
тут ведь должна быть каждая номенклатура! а не список
---
Ну дак сделайте запрос по РН Продажи отберите различную номенклатуру выгрузите в ТЗ, строчки номенклатуры из ТЗ перебирая циклом загоняйте в тот запрос что я выше написал.

Либо раскуривайте запросом тут последнее сообщение
---
```1c
    ВЫБРАТЬ РАЗЛИЧНЫЕ
    ТоварыНаСкладах.Номенклатура КАК Номенклатура
    ПОМЕСТИТЬ Вт1
    ИЗ
    РегистрНакопления.ТоварыНаСкладах КАК ТоварыНаСкладах
    ГДЕ
    ТоварыНаСкладах.Период МЕЖДУ &Дата И &Дата1
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ РАЗЛИЧНЫЕ ПЕРВЫЕ 3
    ТоварыНаСкладах.Период КАК Период,
    ТоварыНаСкладах.Регистратор КАК Регистратор,
    ТоварыНаСкладах.Номенклатура КАК Номенклатура
    ПОМЕСТИТЬ Вт2
    ИЗ
    РегистрНакопления.ТоварыНаСкладах КАК ТоварыНаСкладах
    ГДЕ
    ТоварыНаСкладах.Номенклатура = &Номенклатура
    И ТИПЗНАЧЕНИЯ(ТоварыНаСкладах.Регистратор) = ТИП(Документ.ПоступлениеТоваров)

    СГРУППИРОВАТЬ ПО
    ТоварыНаСкладах.Регистратор,
    ТоварыНаСкладах.Период,
    ТоварыНаСкладах.Номенклатура
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    Вт1.Номенклатура КАК Номенклатура,
    Вт2.Регистратор КАК Регистратор
    ИЗ
    Вт1 КАК Вт1
    ЛЕВОЕ СОЕДИНЕНИЕ Вт2 КАК Вт2
    ПО Вт1.Номенклатура.Ссылка = Вт2.Номенклатура.Ссылка
```
---
как то так примерно
---
Не задано значение параметра "Номенклатура"
ЗакупкиОбороты.Номенклатура = <<?>>&Номенклатура
---
И ТИПЗНАЧЕНИЯ(ТоварыНаСкладах.Регистратор) = ТИП(Документ.ПоступлениеТоваров)
вставил в УПП код (закомментарил указанную строку) и выводит просто список номенклатуры но без Регистратора совсем
---
Сделал запрос по документам, но можно переделать к РН.

```1c
    ВЫБРАТЬ
    Номенклатура.Ссылка КАК Ссылка
    ПОМЕСТИТЬ ВыбраннаяНоменклатура
    ИЗ
    Справочник.Номенклатура КАК Номенклатура
    ГДЕ
    Номенклатура.Ссылка В(&МассивНоменклатуры)
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ВыбраннаяНоменклатура.Ссылка КАК Номенклатура,
    ПоступлениеТоваровУслугТовары.Ссылка КАК Ссылка,
    ПоступлениеТоваровУслугТовары.Ссылка.Дата КАК Дата
    ПОМЕСТИТЬ ПоступленияНоменклатуры
    ИЗ
    ВыбраннаяНоменклатура КАК ВыбраннаяНоменклатура
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ Документ.ПоступлениеТоваровУслуг.Товары КАК ПоступлениеТоваровУслугТовары
    ПО ВыбраннаяНоменклатура.Ссылка = ПоступлениеТоваровУслугТовары.Номенклатура
    И (ПоступлениеТоваровУслугТовары.Ссылка.Проведен)

    СГРУППИРОВАТЬ ПО
    ВыбраннаяНоменклатура.Ссылка,
    ПоступлениеТоваровУслугТовары.Ссылка,
    ПоступлениеТоваровУслугТовары.Ссылка.Дата
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ВложенныйЗапрос.Номенклатура КАК Номенклатура,
    МИНИМУМ(ВложенныйЗапрос.Дата) КАК Дата
    ПОМЕСТИТЬ МинимальныеДатыПоступлений
    ИЗ
    (ВЫБРАТЬ
    Т1.Номенклатура КАК Номенклатура,
    Т1.Дата КАК Дата
    ИЗ
    ПоступленияНоменклатуры КАК Т1
    ЛЕВОЕ СОЕДИНЕНИЕ ПоступленияНоменклатуры КАК Т2
    ПО Т1.Номенклатура = Т2.Номенклатура
    И Т1.Дата <= Т2.Дата

    СГРУППИРОВАТЬ ПО
    Т1.Номенклатура,
    Т1.Дата

    ИМЕЮЩИЕ
    КОЛИЧЕСТВО(РАЗЛИЧНЫЕ Т2.Ссылка) <= 3) КАК ВложенныйЗапрос

    СГРУППИРОВАТЬ ПО
    ВложенныйЗапрос.Номенклатура
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    МинимальныеДатыПоступлений.Номенклатура КАК Номенклатура,
    ПоступленияНоменклатуры.Ссылка КАК Ссылка
    ИЗ
    МинимальныеДатыПоступлений КАК МинимальныеДатыПоступлений
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ ПоступленияНоменклатуры КАК ПоступленияНоменклатуры
    ПО МинимальныеДатыПоступлений.Номенклатура = ПоступленияНоменклатуры.Номенклатура
    И МинимальныеДатыПоступлений.Дата <= ПоступленияНоменклатуры.Дата
```